

PHẦN MỀM MAPLE 9.5 VỚI ỨNG DỤNG TRONG DẠY VÀ HỌC TOÁN HÌNH HỌC GIẢI TÍCH TRONG KHÔNG GIAN.

Với gói lệnh: **> with(geom3d):**

I. ĐIỂM

+Cú pháp: **> point(A, Px, Py, Pz);**

Hoặc **> point(A, [Px, Py, Pz]);**

Là lệnh dựng điểm A có tọa độ Px, Py, Pz.

**** Điểm có tọa độ được chọn ngẫu nhiên:**

Cú pháp1: **> randpoint(name, range1, range2, range3);**

Trong đó: - name : là tên của điểm được dựng

- range1; range2; range3 theo thứ tự là các miền giá trị của hoành độ, tung độ và cao độ của điểm.

Cú pháp02: **> randpoint(name, obj, range1, range2, range3);**

Trong đó: - name : là tên của điểm được dựng

- range1; range2; range3 theo thứ tự là các miền giá trị của hoành độ, tung độ và cao độ của điểm.

- obj: là đường thẳng, mặt phẳng hoặc mặt cầu.

Lệnh thứ 2 để chọn điểm ngẫu nhiên trên đường thẳng, mặt phẳng. mặt cầu !

II. ĐOẠN THẲNG.

Cú pháp: **> segment(seg, [P1, P2]);**

> segment(seg, P1, P2);

*** Đoạn thẳng có định hướng.**

> dsegment(dseg, [P1, P2]);

> dsegment(dseg, P1, P2);

Trong đó: -seg: là tên của đoạn thẳng;

- P1, P2: là tên của hai điểm mút.

*** Trung điểm M của đoạn thẳng AB:**

Cú pháp: **> midpoint(M, A, B);**

hoặc **> midpoint(M, seg);**

Trong đó: - seg là tên của đoạn thẳng đã được xác định trước.

Ví dụ: Cho hai điểm $A(1;2;-2)$, $B(2;3;0)$.

Ta xác định đoạn thẳng AB và đặt tên là AB bằng lệnh sau:

> with(geom3d):

> point(A, 1, 2, -2): point(B, 2, 3, 0):

> segment(AB, A, B);

AB

Dùng hàm **> form(AB);** để xem thể loại của đối tượng vừa định nghĩa:

> form(AB);

segment3d

Kết quả là *segment3d*, cho ta biết đối tượng AB vừa định nghĩa là đoạn thẳng.

Để xác định trung điểm M của đoạn AB nói trên ta dùng lệnh:

```
> midpoint(M,A,B);    #hoac midpoint(M,AB)
M
```

Để xem tọa độ điểm M ta dùng lệnh:

```
> coordinates(M);

$$\left[ \frac{3}{2}, \frac{5}{2}, -1 \right]$$

```

II. ĐƯỜNG THẲNG

1) Đường thẳng đi qua hai điểm A và B cho trước.

+ Cú pháp: **> line(l, [A, B]);**

Trong đó: - A, B là hai điểm đường thẳng đi qua
- l là tên của đường thẳng.

2) Đường thẳng đi qua một điểm A và vuông góc với mặt phẳng p.

+ Cú pháp: **> line(l, [A, p]);**

Trong đó: - A là điểm đường thẳng đi qua, p là mặt phẳng vuông góc với đường thẳng l.
- l là tên của đường thẳng

3) Đường thẳng là giao tuyến của hai mp p1 và p2.

+ Cú pháp: **> line(l, [p1, p2]);**

4) Đường thẳng đi qua một điểm A và có vector chỉ phương v.

+ Cú pháp: **> line(l, [A, v]);**

5) Đường thẳng cho bởi phương trình tham số.

+ Cú pháp: **> line(l, [x0+a*t, y0+b*t, z0+c*t], t);**

Các ví dụ:

Ví dụ 1: Đường thẳng đi qua hai điểm A(1;4;5), B(-4;0;1).

```
> with(geom3d):
Warning, the assigned name polar now has a global binding
> line(d,[point(A,1,4,5),point(B,-4,0,1)]):
Equation(d,`t`);

$$[1-5t, 4-4t, 5-4t]$$

```

Ví dụ 2: Trong không gian với hệ tọa độ Oxyz cho 3 điểm

A(1;3;2), B(1;2;1), C(1;1;1). Hãy viết PTTS của đường thẳng (d) đi qua trọng tâm tam giác ABC và vuông góc với mặt phẳng chứa tam giác ? (ĐH Huế 1999)

Bước 1: Xác định trọng tâm G của tam giác ABC.

```
> with(geom3d):
Warning, the assigned name polar now has a global binding
> triangle(ABC,[point(A,1,3,2),point(B,1,2,1),point(C,1,1,1)]):
centroid(G,ABC):coordinates(G);
```

$$\begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{pmatrix} 1 \\ 2 \\ \frac{4}{3} \end{pmatrix} + t \begin{pmatrix} 1 \\ 2 \\ \frac{4}{3} \end{pmatrix}$$

Giải thích: lệnh **centroid(G,ABC)** xác định trọng tâm của ΔABC .

Bước 2: Dựng mp(p) qua 3 điểm A, B, C.

```
> plane(p,[A,B,C]):Equation(p,[x,y,z]);
```

$$1 - x = 0$$

Bước 3: Dựng đường thẳng (d) thỏa yêu cầu (đi qua G và vuông góc với mặt phẳng p).

```
> line(d,[G,p]):Equation(d,`t`);
```

$$\begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{pmatrix} 1 \\ 2 \\ \frac{4}{3} \end{pmatrix} - t \begin{pmatrix} 1 \\ 2 \\ \frac{4}{3} \end{pmatrix}$$

Ví dụ 3: Cho mp(p): $x + y - z + 1 = 0$ và hai đường thẳng

$$(d1): \begin{cases} x + 2y = 0 \\ y - z + 1 = 0 \end{cases}; \quad (d2): \begin{cases} x - z + 2 = 0 \\ 3y - z + 12 = 0 \end{cases}$$

Gọi (l1), (l2) lần lượt là hình chiếu của (d1), (d2) trên mp(p). Tìm tọa độ giao điểm H của hai đường thẳng (l1), (l2).

(ĐH QG TP HỒ CHÍ MINH_1998)

Bước 1: Xác định các đường thẳng (l1), (l2) nhờ lệnh : >

```
projection(l1,d1,p)
```

```
> line(d1,[plane(p1,x+2*y=0,[x,y,z]),plane(p2,y-z+1=0,[x,y,z])]):
line(d2,[plane(p3,x-z+2=0,[x,y,z]),plane(p4,3*y-z+12=0,[x,y,z])]):
plane(p,x+y-z+1=0,[x,y,z]):
projection(l1,d1,p):projection(l2,d2,p):Equation(l1,`t`);Equation(l2,`u`);
```

$$\begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{pmatrix} \frac{4}{3} \\ -\frac{4}{3} \\ -\frac{5}{3} \end{pmatrix} + t \begin{pmatrix} \frac{5}{3} \\ \frac{2}{3} \\ \frac{1}{3} \end{pmatrix} + u \begin{pmatrix} \frac{1}{3} \\ \frac{2}{3} \\ \frac{1}{3} \end{pmatrix}$$

$$\begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{pmatrix} \frac{1}{3} \\ -\frac{1}{3} \\ -\frac{8}{3} \end{pmatrix} + u \begin{pmatrix} \frac{8}{3} \\ \frac{7}{3} \\ \frac{2}{3} \end{pmatrix} + v \begin{pmatrix} \frac{5}{3} \\ \frac{10}{3} \\ \frac{10}{3} \end{pmatrix}$$

Bước 2: Xác định giao điểm H của (l1), (l2) nhờ lệnh : >

```
intersection(H,l1,l2);
```

```
> intersection(H,l1,l2): H:=coordinates(H);
```

$$H := \begin{pmatrix} \frac{3}{2} \\ -\frac{15}{8} \\ \frac{5}{8} \end{pmatrix}$$

Ví dụ 4:

Trong không gian Oxyz cho điểm $A(0;1;1)$ và hai đường thẳng

$$(d1): \frac{x-1}{3} = \frac{y-2}{1} = \frac{z}{1}; \quad (d2): \begin{cases} x + y - z + 2 = 0 \\ x + 1 = 0 \end{cases}$$

Viết PT đường thẳng (d) qua A, vuông góc với (d1) và cắt (d2).

Chú ý: khi nhập PT (d1) trong Maple, ta nhập ở dạng tham số hoặc dạng 1 (đi qua một điểm và có vectơ chỉ phương)

Bước 1: Xác định mp(p) qua A và vuông góc với (d1):

```
> with(geom3d):
Warning, the assigned name polar now has a global binding
> line(d1,[1+3*t,2+t,t],t):point(A,0,1,1):plane(p,[A,[3,1,1]]):
Equation(p,[x,y,z]);
```

$$-2 + 3x + y + z = 0$$

Bước 2: Xác định mp(q) qua A và chứa (d2)

```
> plane(p1,x+y-z+2=0,[x,y,z]):plane(p2,x+1=0,[x,y,z]):
line(l,[p1,p2]):randpoint(B,1):point(A,0,1,1):line(l1,[A,B]):
plane(q,[l,l1]):Equation(q,[x,y,z]);
```

$$-x + y - z = 0$$

Qua đó, xác định được đường thẳng (d) cần dựng là giao của hai mặt phẳng trên.

Phương trình tham số:

```
> line(d,[p,q]): Equation(d,`t`);
```

$$\begin{cases} \frac{1}{2} - 2t, \frac{1}{2} + 2t, 4t \end{cases}$$

III. MẶT PHẪNG.

1) Dựng mặt phẳng đi qua một điểm A và có vectơ pháp tuyến v cho trước.

Trước tiên ta dùng gói lệnh: **> with(geom3d):**

+Cú pháp: **> plane(p, [A, v]);**

2) Dựng mặt phẳng đi qua ba điểm không thẳng hàng A, B, C.

+Cú pháp: **> plane(p, [A,B,C]);**

3) Dựng mặt phẳng chứa đường thẳng l_1 và song song với đường thẳng l_2 chéo nhau.

+Cú pháp: **> plane(p, [l1, l2]);**

Chú ý: - Nếu nhập l_2 trước thì mp(p) sẽ chứa đường thẳng l_2 và song song với đường thẳng l_1 .

- Nếu 2 đường thẳng l_1, l_2 song song hoặc cắt nhau thì mp(p) sẽ chứa hai đường thẳng đó.

4) Dựng mặt phẳng đi qua điểm A và song song với hai đường thẳng chéo nhau l_1, l_2 .

+Cú pháp: **> plane(p, [A, l1, l2]);**

5) Dựng mặt phẳng theo PTTQ của nó.

+Cú pháp: **> plane(p, equ, list);**

Trong đó: **equ**: là phương trình của mp(p)

list: là danh sách các biến của mp(p), thường là [x,y,z].

Các ví dụ:

Ví dụ 1: Viết PTTQ của mặt phẳng đi qua 3 điểm $A(1;2;3), B(0;-1;1), C(-3;0;2)$.

```
[> with(geom3d):
Warning, the assigned name polar now has a global
binding
[> plane(p,[point(A,1,2,3),point(B,0,-1,1),point(C,-3,0,2)]):
Equation(p,[x,y,z]);
```

$$17 - x + 7y - 10z = 0$$

Ví dụ 2: Viết phương trình tổng quát của mp(p) đi qua điểm $A(1,2,-3)$ và song

song với hai đường thẳng chéo nhau $l_1: \begin{cases} x - y + 3z + 1 = 0 \\ 3x - 2y - z + 2 = 0 \end{cases}$ và $l_2: \begin{cases} x = 3 - 5t \\ y = -2 + t \\ z = 1 + 3t \end{cases}$.

```
> plane(p1,x-y+3*z+1=0,[x,y,z]):plane(p2,3*x-2*y-z+2=0,[x,y,z]):
line(l1,[p1,p2]): line(l2,[3-5*t,2+t,1+3*t],t):point(A,1,2,-3):
plane(p,[A,l1,l2]):Equation(p,[x,y,z]);
```

$$194 + 29x - 26y + 57z = 0$$

Ví dụ 3: Viết PTTQ của mp(p) đi qua điểm $A(1;3;2)$ và chứa đường thẳng (l) :

$$\begin{cases} x - 2y + 3 = 0 \\ 2x + 3z = 0 \end{cases}.$$

```
> with(geom3d):
> plane(p1,x-2*y+3=0,[x,y,z]):plane(p2,2*x+3*z=0,[x,y,z]):
line(l,[p1,p2]):Equation(l,`t`);
```

$$\vec{e} = -6t, \frac{3}{2} - 3t, 4t$$

```
> randpoint(B,l): {lấy điểm B ngẫu nhiên trên đường thẳng (l)}
> point(A,1,3,2):line(l1,[A,B]):Equation(l1,`t`):
{dựng đường thẳng l1 đi qua A và B}
> plane(q,[l,l1]):Equation(q,[x,y,z]); (mp(p) là mp chứa l và l1 _cắt nhau)
```

$$24 + 12x - 16y + 6z = 0$$

Đoạn lệnh chung để dẫn đến kết quả:

```
> plane(p1,x-2*y+3=0,[x,y,z]):
plane(p2,2*x+3*z=0,[x,y,z]):line(l,[p1,p2]):
randpoint(B,l):point(A,1,3,2):line(l1,[A,B]):
plane(q,[l,l1]):Equation(q,[x,y,z]);
```

$$24 + 12x - 16y + 6z = 0$$

Ví dụ 4: (Bạn đọc tự giải)

Viết phương trình của mp(q) đi qua điểm $B(0;-1;5)$ và chứa đường thẳng

$$(d): \begin{cases} x = 2 - 3t \\ y = -3 + t \\ z = 1 - 2t \end{cases}$$

Ví dụ 5: Viết PT mp(p) chứa đường thẳng $(d1): \begin{cases} x = 2 + 2t \\ y = -1 + t \\ z = 1 \end{cases}$ và song song với đường

thẳng $(d2): x = 1; y = 1 + t; z = 3 - t$. (ĐH Huế_1998).

```
> restart;
> with(geom3d):
line(d1,[2+2*t,-1+t,1],t):line(d2,[1,1+t,3-t],t):
plane(p,[d1,d2]): Equation(p,[x,y,z]);
Warning, the assigned name polar now has a global binding
2-x+2y+2z=0
```

Ví dụ 6: (Bạn đọc tự giải)

Viết phương trình mp(q) và mp(p) song song với nhau và lần lượt chứa hai đường thẳng

$$(d1): \begin{cases} x - 8z + 23 = 0 \\ y - 4z + 10 = 0 \end{cases}; (d2): \begin{cases} x - 2z - 3 = 0 \\ y + 2z + 2 = 0 \end{cases}$$

(ĐH KT&T-1995)

Ví dụ 7:

IV. TAM GIÁC.

1. Dựng tam giác có ba đỉnh là A, B, C.

Cú pháp: **> triangle(T, [A, B, C], n)**

- Trong đó: T là tên của tam giác; n là danh sách tên các trục tọa độ (thường $n = [x, y, z]$)
[A, B, C] là tên của ba đỉnh.

2. Dựng tam giác xác định bởi phương trình của ba cạnh.

Cú pháp: **> triangle(T, [l1, l2, l3], n)**

- Trong đó: T là tên của tam giác; n là danh sách tên các trục tọa độ (thường $n = [x, y, z]$)
[l1, l2, l3] là tên của ba cạnh (xác định bởi các phương trình đã định trước).

V. MẶT CẦU

1. Mặt cầu đi qua bốn điểm không đồng phẳng A, B, C, D.

Cú pháp: **> sphere(s, [A, B, C, D], n, 'centername'=m)**

Trong đó: - s: là tên của mặt cầu
- n là danh sách tên các trục tọa độ (thường $n = [x, y, z]$)
- m là tên của tâm mặt cầu.
- [A, B, C, D] là tên bốn điểm mà mặt cầu đi qua.

2. Mặt cầu có đường kính AB_ A, B là 2 điểm phân biệt.

Cú pháp: **> sphere(s, [A, B], n, 'centername'=m)**

Trong đó:

- s: là tên của mặt cầu
- n là danh sách tên các trục tọa độ (thường $n=[x,y,z]$)
- m là tên của tâm mặt cầu.

3. Mặt cầu có tâm A và bán kính bằng rad .

Cú pháp: **> sphere(s, [A, rad], n, 'centername'=m)**

Trong đó:

- s: là tên của mặt cầu
- n là danh sách tên các trục tọa độ (thường $n=[x,y,z]$)
- m là tên của tâm mặt cầu, ta đặt tên cho tâm để tiện khi gọi nó.
- [A, rad]: A là tọa độ tâm, rad là độ dài bán kính.

Ví dụ: Mặt cầu s có tâm A(1;-1;0) và bán kính rad = 5, có phương trình là:

```
> with(geom3d):
> sphere(s,[point(A,1,-1,0),5],[x,y,z],`centername`=m);
```

s

```
> Equation(s,[x,y,z]);
```

$$-23 + x^2 + y^2 + z^2 - 2x + 2y = 0$$

Ta biết, tâm mặt cầu là A. Nhưng ở trên ta đã đặt lại tên cho tâm của mặt cầu s là m. Ta tính tọa độ của tâm m như sau:

```
> coordinates(m);
```

[1, -1, 0]

3. Mặt cầu có tâm A và tiếp xúc với mặt phẳng p .

Cú pháp: **> sphere(s, [A, p], n, 'centername'=m)**

Trong đó:

- s: là tên của mặt cầu
- n là danh sách tên các trục tọa độ (thường $n=[x,y,z]$)
- m là tên của tâm mặt cầu, ta đặt tên cho tâm để tiện khi gọi nó.
- [A, p]: A là tọa độ tâm, p là mặt phẳng tiếp diện của mặt cầu s.

Ví dụ:

Xét mặt cầu (s) có tâm I(1;2;-2) và tiếp xúc với mặt phẳng (p): $x - 2y + z - 3 = 0$.

Ta sẽ lập PT của mặt cầu, xác định tọa độ tâm và tính bán kính của mặt cầu:

```
> restart;with(geom3d):
Warning, the assigned name polar now has a global binding

> point(A,1,2,-2):plane(p,x-2*y+z-3=0,[x,y,z]):
print(Toa_do_cua_A=coordinates(A));print(PT_mpP=Equation(p,[x,y,
z]));
```

Toa_do_cua_A = [1, 2, -2]

PT_mpP = (x - 2 y + z - 3 = 0)

```
> sphere(s,[A,p],[x,y,z],`centername`=A):
`PT cua mat cau tam A va txuc voi
mp(P):`;eq:=Equation(s,[x,y,z]):
student[completesquare](eq,[x,y,z]);
PT cua mat cau tam A va txuc voi mp(P):
```

$$(z+2)^2 - \frac{32}{3} + (y-2)^2 + (x-1)^2 = 0$$

> **Tam của mat cau:** `coordinates(A);`

Ban kinh mat cau: `r:= radius(s);`

Tam của mat cau:

[1, 2, -2]

Ban kinh mat cau:

$$r := \frac{4\sqrt{6}}{3}$$

4. Mặt cầu xác định bởi phương trình .

Cú pháp: `> sphere(s, eqn, n, 'centername'=m)`

5. Các yếu tố khác liên quan đến mặt cầu.

a) Diện tích của mặt cầu s.

Cú pháp: `> area(s);`

b) Mặt phẳng tiếp diện p của mặt cầu s tại điểm A.

Cú pháp: `> TangentPlane(p,A,s);`

c) Thể tích của mặt cầu s.

Cú pháp: `> volume(s);`

d) Phương tích của một điểm M đối với mặt cầu s.

Cú pháp: `> powerps(M,s);`

e) Mặt phẳng đẳng phương p của hai mặt cầu s1 và s2.

Cú pháp: `> RadicalPlane(p, s1 ,s2);`

f) Kiểm tra xem mp(p) có tiếp xúc với mặt cầu (s) hay không

Cú pháp: `> IsTangent(p,s);`

Trường hợp PT mp(p) có chứa tham số, ta dùng lệnh sau để tìm điều kiện của tham số để mp(p) tiếp xúc với mặt cầu (s): `> IsTangent(p,s,'condition'):`

condition;

g) Tính bán kính của mặt cầu s.

Cú pháp: `> radius(s);`

h) Tìm tâm của mặt cầu s.

Cú pháp: `> center(s);`

Các ví dụ:

Trong không gian với hệ trục tọa độ Đề các vuông góc Oxyz cho mặt cầu (S) có phương trình :

$$x^2 + y^2 + z^2 - 2x - 4y - 6z = 0$$

1) Xác định tọa độ tâm và tính bán kính của mặt cầu (S).

2) Xét vị trí tương đối của mặt cầu (S) với mặt phẳng (P): $x + y - z + k = 0$ (k là tham số)

3) Tìm tọa độ giao điểm của (S) với đường thẳng (l) đi qua hai điểm $A(1;1;1)$, $B(2;-1;5)$ và viết phương trình các mặt phẳng tiếp diện của (S) tại các giao điểm đó.

Hướng dẫn giải:

> **restart:with(geom3d):**

Warning, the assigned name polar now has a global binding

> **sphere(s,x^2+y^2+z^2-2*x-4*y-6*z=0,[x,y,z],'centername'=m):plane(p,x+y-z+k=0,[x,y,z]):**
`Toa do tam m cua mat cau (s):`;coordinates(m);`Ban kinh cua mat cau (s):`; r:=radius(s);

Toa do tam m cua mat cau (s):

$[1, 2, 3]$

Ban kinh cua mat cau (s):

$r := \sqrt{14}$

> **`PT mat phang p:`;Equation(p,[x,y,z]);**

PT mat phang p:

$x + y - z + k = 0$

> **`Khoang cach tu tam m den mp(p):`;d:=distance(m,p);**

Khoang cach tu tam m den mp(p):

$d := \frac{1}{3}|k|\sqrt{3}$

> **`mp(p) tiep xuc voi mat cau s <=> d=r:`; `Hay`; solve(d=r, {k});**

mp(p) tiep xuc voi mat cau s <=> d=r:

Hay

$\{k = \sqrt{14}\sqrt{3}\}, \{k = -\sqrt{14}\sqrt{3}\}$

> **`mp(p) cat mat cau (s) <=> d<r:`; `Hay`;solve(d<r, {k});**

mp(p) cat mat cau (s) <=> d<r:

Hay

$\{k < \sqrt{14}\sqrt{3}, -\sqrt{14}\sqrt{3} < k\}$

> **`mp(p) khong cat mat cau (s) <=> d>r:`; `Hay`; solve(d>r, {k});**

mp(p) khong cat mat cau (s) <=> d>r:

Hay

$\{\sqrt{14}\sqrt{3} < k\}, \{k < -\sqrt{14}\sqrt{3}\}$

3)

```
> line(1,[point(A,1,1,1),point(B,2,-1,5)]):
`PTTS của đường thẳng đi qua hai điểm A, B:`;Equation(1,t);
    PTTS của đường thẳng đi qua hai điểm A, B:
        [1+t, 1-2t, 1+4t]

> intersection(obj,1,s):`Toa do cac giao diem cua đường thẳng 1
vói mc(s) là nghiệm của hệ gồm hai phương trình:`;
Equation(1,t);Equation(s,[x,y,z]); `Giải hệ phương trình trên ta
được hai giao
diem:`;M:=coordinates(1_intersect1_s);N:=coordinates(1_intersect
2_s);
areinterls: "two points of intersection"
    Toa do cac giao diem cua đường thẳng 1 với mc(s) là nghiệm của hệ gồm hai phương
    trình:
        [1+t, 1-2t, 1+4t]
        
$$x^2 + y^2 + z^2 - 2x - 4y - 6z = 0$$

        Giải hệ phương trình trên ta được hai giao diem:
            M := [2, -1, 5]
            N :=  $\left[\frac{4}{7}, \frac{13}{7}, \frac{-5}{7}\right]$ 

> `PT mặt tiếp diện của mc(s) tại điểm
M:`;TangentPlane(p1,1_intersect1_s,s):sort(Equation(p1,[x,y,z]),
[x,y,z]);
    PT mặt tiếp diện của mc(s) tại điểm M:
        
$$-x + 3y - 2z + 15 = 0$$


> `PT mặt tiếp diện của mc(s) tại điểm N:`;
TangentPlane(p1,1_intersect2_s,s):
P:=sort(Equation(p1,[x,y,z]),[x,y,z]);`Hay`;primpart(lhs(P),x)=0
;
    PT mặt tiếp diện của mc(s) tại điểm N:
        
$$P := \frac{3x}{7} + \frac{y}{7} + \frac{26z}{7} + \frac{15}{7} = 0$$

        Hay
        
$$3x + y + 26z + 15 = 0$$

```

Ví dụ 2:

Trong không gian với hệ trục tọa độ Đề các vuông góc Oxyz cho bốn điểm $A(6;-2;3)$, $B(0;1;6)$, $C(2;0;-1)$, $D(4;1;0)$.

- 1) Chứng minh rằng A, B, C, D là bốn đỉnh của một tứ diện.
- 2) Tính thể tích của tứ diện $ABCD$.
- 3) Viết phương trình của mặt cầu ngoại tiếp tứ diện $ABCD$. Xác định tọa độ tâm và tính bán kính của mặt cầu đó.

4) Xác định tọa độ tâm và tính bán kính của đường tròn đi qua ba điểm A, B, C .

Hướng dẫn:

> **restart:**

> **with(geom3d):**

Warning, the assigned name polar now has a global binding

> **point(A,6,-2,3):point(B,0,1,6):point(C,2,0,-1):point(D,4,1,0):**

> **AreCoplanar(A,B,C,D);**

false

☞ Lệnh trên cho kết quả *false* chứng tỏ 4 điểm A, B, C, D không đồng phẳng. Vậy chúng là bốn đỉnh của một tứ diện.

> **gtetrahedron(T, [A, B, C, D]):`The tích của tu dien**

ABCD: `;V:=volume(T);

The tích của tu dien ABCD:

V := 12

☞ Lệnh **gtetrahedron(T, [A, B, C, D])** để dựng tứ diện $ABCD$.

> **sphere(s,[A, B, C, D], [x,y,z], 'centername'=m):`PT mat cau di qua 4 diem A, B, C, D: `;**

student[completesquare](Equation(s,[x,y,z]),[x,y,z]);

PT mat cau di qua 4 diem A, B, C, D:

$$(z-3)^2 - 17 + (y+1)^2 + (x-2)^2 = 0$$

> **`Toa do tam của mat cau: `;coordinates(m);**

Toa do tam của mat cau:

[2, -1, 3]

> **`Ban kinh của mat cau: `;r:=radius(s);**

Ban kinh của mat cau:

$$r := \sqrt{17}$$

☞ Trong Maple chưa có lệnh xác định giao tuyến của mặt phẳng với mặt cầu. Vậy để tính bán kính của đường tròn giao tuyến ta tính theo công thức

$R = \sqrt{r^2 - \left[d_{(m,(ABC))} \right]^2}$ – trong đó $d_{(m,(ABC))}$ là khoảng cách từ tâm m của $mc(s)$ đến $mp(ABC)$.

> **`PTTQ của mp(ABC): `; plane(ABC,[A,B,C]):**

Eq:=Equation(ABC,[x,y,z]): primpart(lhs(Eq))=0;

PTTQ của mp(ABC):

$$-x - 2y + 2 = 0$$

> `Khoang cach tu tam m cua mc(s) den
mp(ABC):` ;d:=distance(m,ABC);

Khoang cach tu tam m cua mc(s) den mp(ABC):

$$d := \frac{2\sqrt{5}}{5}$$

> `Ban kinh cua duong tron di qua ba diem A, B, C:`;
R:=sqrt(r^2-d^2);

Ban kinh cua duong tron di qua ba diem A, B, C:

$$R := \frac{9\sqrt{5}}{5}$$

> line(l,[m,ABC]):`PTTS cua duong thang d di qua tam m cua mc(S)
va vuong goc voi mp(ABC):`;Equation(l,t);`Tam cua duong tron di
qua ba diem A, B, C la giao diem cua mp(ABC) va dt (d)`;`Nen
toa do tam duong tron la nghiem cua he gom 2
PT:`;Equation(l,t);Equation(s,[x,y,z]);`Toa do tam cua duong
tron di qua ba diem A, B, C:`; intersection(O,l,ABC):
coordinates(O);

PTTS cua duong thang d di qua tam m cua mc(S) va vuong goc voi mp(ABC):

$$[2 - 18t, -1 - 36t, 3]$$

Tam cua duong tron di qua ba diem A, B, C la giao diem cua mp(ABC) va dt (d)

Nen toa do tam duong tron la nghiem cua he gom 2 PT:

$$[2 - 18t, -1 - 36t, 3]$$

$$-3 + x^2 + y^2 + z^2 - 4x + 2y - 6z = 0$$

Toa do tam cua duong tron di qua ba diem A, B, C:

$$\left[\frac{12}{5}, \frac{-1}{5}, 3 \right]$$

VI. CÁC PHÉP DỜI HÌNH TRONG KHÔNG GIAN.

1. Phép tịnh tiến

Xác định ảnh H1 của hình H qua phép tịnh tiến theo vector \overline{AB}

Cú pháp: > translation(H1, H, AB);

Trong đó AB là đoạn thẳng định hướng.

2. Phép đối xứng

a) qua một mặt phẳng

Xác định ảnh H1 đối xứng của hình H qua mặt phẳng (P)

Cú pháp: > reflection(H1, H, P);

b) qua một đường thẳng

Xác định ảnh H1 đối xứng của hình H qua đường thẳng l .

Cú pháp: **> reflection(H1, H, l);**

c) qua một điểm

Xác định ảnh H1 đối xứng của hình H qua điểm O

Cú pháp: **> reflection(H1, H, O);**

3. Phép quay

Xác định ảnh H1 của hình H qua phép quay ${}^{\alpha}_lQ$ có trục quay là đường thẳng l và góc quay α .

Cú pháp: **> rotation(H1, H, α , l);**

Trong đó

4. Phép vị tự

Xác định ảnh H1 của hình H qua phép vị tự tâm O tỉ số k_{OV} .

Cú pháp: **> homothety(H1, H, k, O);**

Trong đó

5. Phép đối xứng trượt

Xác định ảnh H1 của hình H qua phép đối xứng trượt gồm phép đối xứng qua mp(P) và phép tịnh tiến theo vector \overline{AB} .

Cú pháp: **> rotation(H1, H, P, AB);**

Trong đó AB là đoạn thẳng định hướng.

6. Phép chiếu vuông góc

a) Xác định điểm A1 là hình chiếu của điểm A trên đường thẳng l hoặc mp(P).

Cú pháp: **> rotation(A1, A, l);** hoặc **> rotation(A1, A, l);**

b) Xác định hình chiếu $l1$ của đường(đoạn) thẳng l trên mp(P).

Cú pháp: **> rotation(l1, l, P);**

Biên soạn: ĐỖ CAO LONG

Địa chỉ: Cụm 3/2, thị trấn Khe Tre, huyện Nam Đông, tỉnh T.T.Huế.

Mail: dclnamdong@yahoo.com.vn

Tel: 054 875045 or 0982013906

Một số tính năng khác

DỤNG

Tìm giao của các đối tượng hình học trong không gian

Cú pháp:

```
> intersection(obj, l1, l2) (1)
> intersection(obj, p1, p2) (2)
> intersection(obj, l1, p1) (3)
> intersection(obj, l1, s) (4)
> intersection(obj, p1, p2, p3) (5)
```

* Lệnh (1) tìm giao điểm của hai đường thẳng, kết quả là một “điểm” nếu hai đường thẳng cắt nhau, ngược lại kết quả là “Null”.

Lệnh (2) tìm giao tuyến của hai mặt phẳng, kết quả là một “đường thẳng” nếu hai mặt phẳng cắt nhau, ngược lại kết quả là “Null”.

Lệnh (3) tìm giao điểm của đường thẳng và mặt phẳng, kết quả là một “điểm” nếu đường thẳng và mặt phẳng cắt nhau, ngược lại kết quả là “Null”.

Lệnh (4) tìm giao điểm của đường thẳng và mặt cầu

Lệnh (5) tìm giao điểm của ba mặt phẳng

Còn **obj** là tên của đối tượng dựng được.

Kiểm tra xem ba điểm A, B, C có thẳng hàng hay không

Cú pháp: > **AreCollinear(A,B,C);**

Kiểm tra xem các đối tượng có đồng phẳng hay không

Kiểm tra xem 4 điểm A, B, C, D có đồng phẳng hay không

Cú pháp:

```
> with(geom3d):
> AreCoplanar(A,B,C,E);
```

Kiểm tra xem hai đường thẳng l1, l2 có đồng phẳng hay không

Cú pháp:

```
> with(geom3d):
> AreCoplanar(l1,l2);
```

Chú ý: Nếu các điểm A, B, C, D hoặc các đường thẳng l1, l2 đồng phẳng đồng phẳng thì kết quả hiển thị là *true*.

TÍNH TOÁN

Góc:

Cú pháp:

```
> FindAngle(l1, l2) (1)
> FindAngle(p1, p2) (2)
> FindAngle(s1, s2) (3)
> FindAngle(l1, p1) (4)
> FindAngle(A, T) (5)
```

* Lệnh (1) tính góc giữa hai đường thẳng l1, l2 nếu chúng cắt nhau

Lệnh (2) tính góc giữa hai mặt phẳng p1, p2

Lệnh (3) tính góc giữa hai mặt cầu cắt nhau

Lệnh (4) tính góc giữa đường thẳng $l1$ và mặt phẳng $p1$.

Lệnh (5) tính góc A của tam giác T , khi A là đỉnh của T .

Khoảng cách:

Khoảng cách giữa hai điểm phân biệt A, B

Cú pháp: **> distance(A,B);**

Khoảng cách giữa hai đường thẳng $l1$ và $l2$

Cú pháp: **> distance(l1,l2);**

Khoảng cách giữa hai mặt phẳng ($p1$) và ($p2$)

Cú pháp: **> distance(p1,p2);**

Khoảng cách từ một điểm A đến một đường thẳng l

Cú pháp: **> distance(A,l);**

Khoảng cách từ một điểm A đến một mặt phẳng (p)

Cú pháp: **> distance(A,p);**

Khoảng cách giữa đường thẳng l và mặt phẳng (p) song song với l

Cú pháp: **> distance(l,p);**

Dựng đường thẳng hoặc mặt phẳng song song

Cú pháp: **> paralell(w,u,v);**

+ w là tên của đường thẳng hoặc mặt phẳng dựng được

+ u là điểm hoặc đường thẳng

+ v là đường thẳng hoặc mặt phẳng

Chú ý:

- Nếu u là điểm còn v là đường thẳng (hoặc mặt phẳng) thì w là đường thẳng (hoặc mặt phẳng) đi qua điểm u và song song với đường thẳng (mặt phẳng) v .

- Nếu u là đường thẳng và v là đường thẳng thì w là mặt phẳng chứa u và song song với v .

II.